FILE monkey.c

% File: monkey.c

:- macros maxtime -> 4.

:- include ’C’; ’monkey.p’.

:- sorts object; location.

:- variables

O :: object;

L, L1, L2 :: location;

A, A1 :: action.

:- constants

monkey, box, banana :: object;

l1, l2, l3 :: location;

at(object,location) :: inertialTrueFluent;

onBox, hasBanana :: inertialFluent;

walk(location) :: action;

pushBox(location) :: action;

climbBox, graspBanana :: action.

nonexecutable A && A1 if -(A=A1).

walk(L) causes at(monkey,L).

nonexecutable walk(L) if at(monkey,L).

nonexecutable walk(L) if onBox.

pushBox(L) causes at(monkey,L).

pushBox(L) causes at(box,L).

nonexecutable pushBox(L) if at(monkey,L).

nonexecutable pushBox(L) if onBox.

nonexecutable pushBox(L) if at(monkey,L1) && at(box,L2) && -(L1=L2).

climbBox causes onBox.

nonexecutable climbBox if onBox.

nonexecutable climbBox if at(monkey,L) && at(box,L1) && -(L=L1).

graspBanana causes hasBanana.

nonexecutable graspBanana if -onBox.

nonexecutable graspBanana if at(monkey,L) && at(banana,L1) && -(L=L1).

caused -at(O,L1) if at(O,L) && -(L=L1).

caused at(banana,L) if at(monkey,L) && hasBanana.

always at(monkey,L) && onBox ->> at(box,L).

FILE monkey.p

%File: monkey.p

:- plan

facts ::

init at(monkey,l1),

init at(banana,l2),

init at(box,l3);

goal ::

last hasBanana.

A DIALOG WITH CCALC

| ?- loadf(’monkey.c’).

% loading file /v/hank/v43/vl/ccalc/monkey.c

% loading file /v/hank/v43/vl/ccalc/C

% loading file /v/hank/v43/vl/ccalc/monkey.p

% 95 atoms, 726 rules, 672 clauses, 1160 msec.

yes

| ?- plan 0.

calling rel_sat...

0. at(banana,l2) at(box,l3) at(monkey,l1)

ACTIONS: walk(l3)

1. at(banana,l2) at(box,l3) at(monkey,l3)

ACTIONS: pushBox(l2)

2. at(banana,l2) at(box,l2) at(monkey,l2)

ACTIONS: climbBox

3. onBox at(banana,l2) at(box,l2) at(monkey,l2)

ACTIONS: graspBanana

4. hasBanana onBox at(banana,l2) at(box,l2) at(monkey,l2)

Elapsed Time (cpu sec): 0

Verify plan? (y/n) y

verifying....

The plan is verified!

yes

| ?-